Generating One-Address Code (5)

using Library;

COMPILER Expr $NC /* token names, generate compiler driver */

/* Convert infix expressions into simple one-address code */

CHARACTERS

  digit  =  "0123456789" .

  letter =  "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz" .

IGNORE CHR(9) .. CHR(13)

TOKENS

  number = digit { digit } .

  variable = letter .

PRODUCTIONS

  Expr

  = { Expression<1> ";"   (. IO.WriteLine(); .)

    } .

  Expression<int R>

  = Term<R>

    {   "+" Term<R+1>    (. IO.WriteLine("ADD R" + R + ",R" + (R+1)); .)

      | "-" Term<R+1>    (. IO.WriteLine("SUB R" + R + ",R" + (R+1)); .)

    } .

  Term<int R>

  = Factor<R>

    {   "*" Factor<R+1>  (. IO.WriteLine("MUL R" + R + ",R" + (R+1)); .)

      | "/" Factor<R+1>  (. IO.WriteLine("DVD R" + R + ",R" + (R+1)); .)

    } .

  Factor<int R>          (. char ch; int n; .)

  =   Identifier<out ch> (. IO.WriteLine("LDA R" + R + "," + ch); .)

    | Number<out n>      (. IO.WriteLine("LDI R" + R + "," + n); .)

    | "(" Expression<R> ")" .

  Identifier<out char ch>

  =  variable            (. ch = token.str[0]; .) .

  Number<out int value>

  =  number              (. try {

                              value = Convert.ToInt32(token.str);

                            } catch (Exception) {

                              value = 0; 
                              SemError("number too large");

                            } .) .

END Expr.

