A simple PVM stack assembler (7)
PRODUCTIONS

  Parva

  = "ASSEM" "BEGIN"

       { Statement }

    "END" "."                    (. Table.CheckLabels(); .) .

  Statement

  = OneWord | TwoWord | WriteString | Label | Branch .

  OneWord

  = (   "ADD"  | "AND"  | "ANEW" | "CEQ"  | "CGE"  | "CGT"  | "CLE"  | "CLT"

      | "CNE"  | "DIV"  | "HALT" | "INPB" | "INPI" | "LDV"  | "LDXA" | "MUL"

      | "NEG"  | "NOP"  | "NOT"  | "OR"   | "PRNB" | "PRNI" | "PRNL" | "REM"

      | "STO"  | "SUB" )         (. CodeGen.OneWord(token.val); .) .

  TwoWord                        (. int value; .)

  = ( "DSP" | "LDC" | "LDA" )    (. string mnemonic = token.val; .)

    SignedNumber<out value>      (. CodeGen.TwoWord(mnemonic, value); .) .

  SignedNumber<out int value>    (. int sign = 1; .)

  = [   "+"

      | "-"                      (. sign = -1; .)

    ] IntConst<out value>        (. value = sign * value; .) .

  IntConst<out int value>

  = number                       (. try {

                                      value = Convert.ToInt32(token.str);

                                    } catch (Exception) {

                                      value = 0; SemError("number too large");

                                    } .) .

  WriteString                    (. string str; .)

  = "PRNS" StringConst<out str>  (. CodeGen.WriteString(str); .) .

  StringConst<out string str>

  = stringLit                    (. str = token.str.Substring(1, 

                                                    token.str.Length-2); .) .

  Label

  = label                        (. string name = token.val.Substring(0,

                                                         token.val.Length-1);

                                    Entry tableEntry = Table.Find(name);

                                    if (tableEntry == null)

                                      Table.Insert(new Entry(name, 

                                                   new Label(known)));

                                    else if (tableEntry.label.IsDefined())

                                      SemError("redefined label");

                                    else tableEntry.label.Here(); .) .

  Branch                         (. int target;

                                    string name;

                                    Label lab; .)

  = ( "BRN" | "BZE" )            (. string mnemonic = token.val; .)

    (   IntConst<out target>     (. CodeGen.TwoWord(mnemonic, target); .)

      | Ident<out name>          (. Entry tableEntry = Table.Find(name);

                                    if (tableEntry == null) {

                                      lab = new Label(!known);

                                      Table.Insert(new Entry(name, lab));

                                    }

                                    else lab = tableEntry.label;

                                    CodeGen.Branch(mnemonic, lab); .)

    ) .

  Ident<out string name>

  = identifier                   (. name = token.val; .) .

